

O Excel não é apenas um software de planilhas eletrônicas, é uma plataforma que possui as mais variadas utilidades, as quais vão desde elaboração de relatórios simples, criação de painéis e análise de dados, até funções mais complexas com aplicações em banco de dados, soluções de engenharia, criação de soluções corporativas, sistemas e as mais variadas funções que variam de acordo com o objetivo do usuário.

Para dominar esta ferramenta, é preciso passar por uma série de etapas, as quais vão desde o aprendizado básico até as funcionalidades mais complexas, para tanto, este livro tem o objetivo de facilitar o aprendizado do leitor em relação as funções. Uma ferramenta do fundamental para ter um maior entendimento das possibilidades que o Excel oferece.

Sumário

INTRODUÇÃO

FUNÇÕES

FUNÇÕES CONDICIONAIS

SE ()

E()

OU()

SEERRO()

FUNÇÕES DE MANIPULAÇÃO DE CADEIA DE TEXTO

ESQUERDA()

EXT.TEXTO()

DIREITA()

LOCALIZAR()

NÚM.CARACT()

SUBSTITUIR ()

COMBINANDO AS FUNÇÕES

TEXTO()

ARRUMAR()

LETRAS MAIÚSCULAS

VERIFICAR CÉLULA

FUNÇÕES DE LOCALIZAÇÃO

ÍNDICE()

CORRESP()

ÍNDICE(CORRESP())

INDICE(CORRESP(); CORRESP())

PROCV()

PROCH()

PROCV(MAIOR())

DESLOC()

DESLOC(CORRESP())

INDIRETO()

CONT.VALORES()

INDICE(CONT.VALORES())

SEERRO() + FUNÇÕES DE PROCURA

FUNÇÕES DE SOMA E MÉDIA

SOMA()

SOMARPRODUTO ()

SUBTOTAL()

SOMASE()

SOMASES()

MEDIA()

MEDIASE()

OUTRAS FUNÇÕES MATEMÁTICAS

MÍNIMO ()

MÁXIMO ()

MENOR ()

MAIOR()

FATORIAL()

ARREDONADAMENTOS

ARRED()

ARREDONDAR.PARA.BAIXO()

ARREDONDAR.PARA.CIMA()

NÚMEROS ALEATÓRIOS

ALEATÓRIO()

ALEATÓRIOENTRE()

CONVERSÃO

CONVERTER()

MUDAR BASE NUMÉRICA

DATA E HORA

DIATRABALHOTOTAL()

FUNÇÕES DE HORÁRIO

FUNÇÕES DE DATA

INTRODUÇÃO

A estrutura deste livro, consiste na apresentação das funções seguida de uma série de exemplos, os quais serão minuciosamente explorados para apresentar de forma clara e objetiva o funcionamento das fórmulas. Para diversos casos, serão apresentadas fórmulas importantes que dependem de uma combinação de outras mais simples, para tanto, estas serão apresentadas inicialmente, para posteriormente demonstrar suas combinações e implicações em resultados poderosos e fascinantes.

Ao longo do livro serão demonstradas inúmeras funções que se encaixam dentro de uma categoria e depois as mesmas poderão reaparecer em combinações com outras, criando assim formulas mais elaboradas as quais podem atingir os mais diversos objetivos. Como por exemplo a categoria manipulação de texto, a qual possui diversas funções simples e depois suas combinações são exploradas para demonstrar seu real potencial.

Este livro não precisa ser necessariamente lido de maneira linear, pode-se utilizado de acordo com a necessidade específica de cada usuário, porém para funções que se encaixam dentro de uma mesma categoria, é recomendado ler todas que pertencem a um mesmo grupo, para facilitar o entendimento de suas combinações.

DEFINIÇÃO

As funções no Excel, também conhecidas como fórmulas, são basicamente instruções matemática que o Excel interpreta de maneira prática para cumprir os mais diversos objetivos. Através delas, pode-se realizar desde os cálculos mais simples até os mais complexos, pode-se também realizar tratamento de dados, manipular cadeias de textos e até mesmo realizar operações lógicas avançadas.

A versatilidade das funções do Excel possibilita realizar diversas tarefas tanto para objetivo pessoal como profissional, abaixo são listado alguns exemplos de possibilidades através da utilização deste incrível recursos:

- Controle de finanças pessoais
- Cálculos de finanças empresarias
- Tratamento e processamento de dados
- Elaborar relatórios de forma rápida e prática
- Criar cálculos sofisticados de engenharia física e matemática

FUNÇÕES CONDICIONAIS

As funções condicionais são primordiais para o controle de informações no Excel, através delas pode-se realizar uma série de análises, estabelecer parâmetros para realizar operações e criar condições para aplicação de outras funções. São funções simples, diretamente ligadas ao significado de seus nomes.

Primeiramente, é preciso apresentar de forma breve os operadores lógicos que serão constantemente utilizados para as funções condicionais:

>	Maior
<	Menor
>=	Maior Igual
<=	Menor Igual
=	Igual
≠	Diferente

SE ()

Esta função é fundamental para o domínio do Excel, basicamente, fornecendo uma condição, pode-se estabelecer um retorno para o caso verdadeiro, e um retorno para o caso falso.

C2							
	A	B	C	D	E	F	G
1	Aluno	Nota	Condicional				
2	Maria	5,1	Reprovado				
3	Pedro	6,0	Aprovado				
4	João	7,4	Aprovado				
5	Flavia	4,1	Reprovado				
6	Maria	3,8	Reprovado				
7	Adriana	8,9	Aprovado				

=SE(B2< 6;"Reprovado";"Aprovado")

B2< 6 - Condicional que será aplicada

Reprovado – Retorno para condição atendida

Aprovado – Retorno para condição não atendida

Uma vez que a condição seja atendida, ou seja, a nota seja inferior ao valor determinado ($B2 < 6$), existirá um retorno para célula, o qual será o primeiro argumento determinado de valor “Reprovado”, caso contrário, ou seja, condição não atendida, o valor retornado será “Aprovado”.

E()

O principal objetivo desta função, é unir dois argumentos ou mais dentro de uma condicional, ela pode ser utilizada isoladamente para retornar verdadeiro ou falso, porém sua utilidade se comprova de fato dentro de outras funções, conforme exemplo a seguir:

	A	B	C	D	E	F	G	H
1	Aluno	Nota	Frequência	Função E				
2	Maria	5,1	85%	=SE(E(B2>=6;				
3	Pedro	6,0	80%	Aprovado				
4	João	7,4	100%	Aprovado				
5	Flavia	4,1	100%	Reprovado				
6	Maria	3,8	90%	Reprovado				
7	Adriana	8,9	60%	Reprovado				

=SE(E($B2 \geq 6$; $C2 > 0,75$); "Aprovado"; "Reprovado")

$B2 \geq 6$ – Primeira condicional

$C2 > 0,75$ – Segunda condicional

Aprovado - Retorno para as duas condicionais atendidas

Reprovado - Retorno para as duas condicionais não atendidas

Neste exemplo, diferente do anterior, a condicional foi invertida, desta vez foi apresentada a condição para nota ser maior ou igual a 6 ($B2 \geq 6$), adicionando desta vez uma segunda condição, $C2 > 0,75$, ou seja, a frequência precisa também ser acima de 75%. Apenas com as duas condicionais atendidas, o retorno será o valor “Aprovado”, caso contrário, o retorno será “Reprovado”.

OU()

Novamente, da mesma maneira que a função E(), a função OU() pode ser utilizada isoladamente, para retornar o valor verdadeiro ou falso, porém sua utilidade se mostra mais presente com a sua utilização dentro da função SE(), conforme exemplo a seguir:

	A	B	C	D	E	F	G	H
1	Aluno	Nota V1	Nota V2	Função Ou				
2	Maria	5,1	7,0	=SE(OU(B2>=6;C2>=6);"Aprovado";"Reprovado")				
3	Pedro	6,0	6,0	Aprovado				
4	João	7,4	7,4	Aprovado				
5	Flavia	4,1	3,6	Reprovado				
6	Maria	3,8	5,0	Reprovado				
7	Adriana	8,9	5,0	Aprovado				

=SE(OU(B2>=6;C2>6);"Aprovado";"Reprovado")

Este exemplo simula uma situação onde um aluno necessita da nota maior que 6 em pelo menos uma das avaliações, desta maneira, utiliza-se a função OU(), com as duas condicionais relacionadas a primeira coluna B2>=6, e depois aplica-se o mesmo critério para segunda coluna C2>6, caso uma das duas seja atendida, o valor retornado será "Aprovado", caso contrário "Reprovado".

SEERRO()

Esta é uma função importantíssima, porém que deve ser utilizada com cautela, uma vez que ela pode ocultar erros que se deseja enxergar. Sua principal aplicação é em fórmulas onde os erros são esperados para determinados casos, porém não deseja-se apresentar os valores com #N/D, #VALOR e etc. Ou para casos, onde caso um erro seja encontrado, realize-se um outro procedimento.

Basicamente o funcionamento, se dá por uma condicional, que caso seja atendida, fará algum outro procedimento, da seguinte maneira:

		<div style="border: 1px solid black; padding: 2px;"> D3 : ✕ ✓ fx =C3/B3 </div>		
	A	B	C	D
1		Custo	Venda	Lucro
2	Departamento	Janeiro	Janeiro	Janeiro
3	Feminino	23.887,00	25.320,22	106%
4	Masculino	16.992,00	16.652,16	98%
5	Infantil	89.002,00	91.672,06	103%
6	Infanto-Juvenil	12.887,00	15.464,40	120%
7	Lingerie		18.818,91	#DIV/0!
8	Calçados Masculinos	5.664,00	5.833,92	103%
9	Calçados Femininos	9.988,00	11.286,44	113%
10	Calçados Infantis		3.209,64	#DIV/0!
11	Acessórios	1.203,00	1.479,69	123%

Observa-se neste exemplo que a coluna D relacionada aos lucros, realiza uma divisão entre os valores de custo pelos valores de venda. Porém caso o valor de custo não seja preenchido, a fórmula aplicada na coluna D, apresentará um erro, pois não se pode dividir 0 por nenhum número. Para contornar este erro, pode-se utilizar a função SEERRO()

	A	B	C	D	E
1		Custo	Venda	Lucro	
2	Departamento	Janeiro	Janeiro	Janeiro	
3	Feminino	23.887,00	25.320,22	106%	
4	Masculino	16.992,00	16.652,16	98%	
5	Infantil	89.002,00	91.672,06	103%	
6	Infanto-Juvenil	12.887,00	15.464,40	120%	
7	Lingerie		18.818,91	Custo não preenchido	
8	Calçados Masculinos	5.664,00	5.833,92	103%	
9	Calçados Femininos	9.988,00	11.286,44	113%	
10	Calçados Infantis		3.209,64	Custo não preenchido	
11	Acessórios	1.203,00	1.479,69	123%	

Desta forma, ao invés de apresentar a mensagem de erro, o texto: “Custo não preenchido” será apresentado.

FUNÇÕES DE MANIPULAÇÃO DE CADEIA DE TEXTO

Existem diversas funções para manipular textos, onde o objetivo é basicamente obter frações do valor de uma célula. São de grande utilidade para usuários que lidam com grandes quantidades de informação, onde muitas vezes se obtém os dados sem formatação, extrações de banco de dados e de demais origens diversas.

As funções ESQUERDA(), DIREITA() e EXT.TEXTO(), por si só já são de extrema utilidade, quando combinadas com LOCALIZAR() e/ou NÚM.CARACT(), podem trazer uma função adicional de grande utilidade para manipular as cadeias de textos.


Primeiramente, vamos apresentar de maneira breve a utilização das funções

ESQUERDA(), DIREITA(), EXT.TEXTO(), LOCALIZAR() e NÚM.CARACT(), para então apresentar as suas combinações.

ESQUERDA()

Esta função é utilizada para se obter o prefixo de um texto.

Exemplo 1:



	A	B	C	D
1				
2	Departamento	Código1	Depto	Código2
3	01 FEM #1	01	FEM	#1
4	02 MAS #2	02	MAS	#2
5	03 INF #3	03	INF	#3

=ESQUERDA(A3;2)

A3 – Célula que será aplicada a função

2 – Significa que será retornado os 2 primeiros caracteres da esquerda

Neste exemplo, o departamento está acompanhado de seu código inicial e final, porém deseja-se separá-los. Para se obter o código inicial (código1), se utiliza a função esquerda, com 2 caracteres.

EXT.TEXTO()

Esta função é utilizada para se extrair a parte central de uma célula.

Exemplo 1:

C3	:	X	✓	<i>fx</i>	=EXT.TEXTO(A3;4;3)
	A	B	C	D	
1					
2	Departamento	Código1	Depto	Código2	
3	01 FEM #1	01	FEM	#1	
4	02 MAS #2	02	MAS	#2	
5	03 INF #3	03	INF	#3	

=EXT.TEXTO(A3; 3;3)

A3 – Célula que será aplicada a função

4 – Significa que será retornado a partir no terceiro caractere

3 – Significa que será retornado 3 caracteres.

Agora deseja-se obter a abreviação FEM, desta forma, se utiliza o EXT.TEXTO(). Nesta função, é preciso informar a posição do caractere inicial e a posição final. Neste exemplo foi utilizado **4**, para inicial, visto que se deseja extrair a partir do quarto caractere (o espaço conta como 1 caractere), por último é informado o número 3, pois deseja-se retornar 3 dígitos.

DIREITA()

Esta função é utilizada para se obter o sufixo de uma célula.

Exemplo:

D3	:	X	✓	<i>f_x</i>	=DIREITA(A3;2)
	A	B	C	D	
1					
2	Departamento	Código1	Depto	Código2	
3	01 FEM #1	01	FEM	#1	
4	02 MAS #2	02	MAS	#2	
5	03 INF #3	03	INF	#3	

=DIREITA(A3;2)

A3 – Célula que será aplicada a função

2 – Significa que será retornado os 2 primeiros caracteres da direita

Esta função tem o mesmo comportamento da ESQUERDA(), porém ao invés de contar a partir do primeiro caractere, ela conta a partir do último caractere, retornando os caracteres a partir da direita.

LOCALIZAR()

Esta função é extremamente simples, o objetivo dela é apenas retornar a posição de um caractere desejado.

Exemplo:

B3	:	X	✓	<i>f_x</i>	=LOCALIZAR(".",A3;1)
	A	B	C	D	
1					
2	Código item	Localizar			
3	D2123044.Camisa_Gola_V	9			
4	D23551.Calça_Jeans44	7			
5	D31002.Calça_Social42	7			

=LOCALIZAR(".;";A3;1)

“.” – Texto que se deseja localizar, neste exemplo deseja-se localizar o ponto.

A3 – Célula que será aplicada a função

1 – A partir de qual caractere será localizado.

Neste exemplo deseja-se buscar o caractere ponto, desta maneira coloca-se entre aspas (“.”), como se deseja localizar a partir do início, a função é finalizada com o número 1. O valor encontrado será 9, pois o ponto se encontra na nona posição.

NÚM.CARACT()

Assim como a anterior, esta função também é extremamente simples, o seu objetivo é simplesmente contar o número total de caracteres de uma célula.

Exemplo:

C3	:	X ✓ fx	=NÚM.CARACT(A3)	
	A	B	C	D
1				
2	Código item	Localizar	Núm.Caract	
3	D2123044.Camisa_Gola_V	9	22	
4	D23551.Calça_Jeans44	7	20	
5	D31002.Calça_Social42	7	21	

=NÚM.CARACT(A3)

A3 – Célula que será aplicada a função

Neste exemplo, a função é aplicada na célula A3 e o valor encontrado é 22, visto que o item tem 22 caracteres.

SUBSTITUIR ()

Este função, apesar de não ter utilidade em combinação com as demais apresentadas, possui uma grande utilidade para tratamento de dados, uma vez que ela pode substituir um termo procurado, por outro informado.

Exemplo 1:

B1		✕ ✓ <i>fx</i>		=SUBSTITUIR(A1;".";",")		
	A	B	C	D	E	F
1	13123.54	13123,54				
2	23123.45	23123,45				
3	123.23	123,23				
4	4.532.135	4532135				
5	3123.56	3123,56				

=SUBSTITUIR(A1;".";",")

A1 – Célula onde a função será aplicada

"." – Texto que será substituído

"," – Texto novo que será implementado

Este exemplo demonstra a aplicação da função substituir para trocar o ponto de uma célula pela virgula, desta forma, sempre que uma virgula for encontrada, ela será substituída por ponto.

Obs: O argumento final de ocorrência não foi declarado por não ter necessidade neste exemplo.

Exemplo 2:

B1		✕ ✓ <i>fx</i>		=SUBSTITUIR(A1;"2";"3";2)		
	A	B	C	D	E	
1	2018 - Trimestre 2	2018 - Trimestre 3				
2						

=SUBSTITUIR(A1;"2";"3";2)

A1 – Célula onde a função será aplicada

"2" – Texto que será substituído

"3" – Texto novo que será implementado

2 – Ocorrência onde a troca será realizada

Neste exemplo, deseja-se substituir o número 2 por 3, porém é preciso informar em qual ocorrência esta troca irá ocorrer, para não alterar erroneamente o ano. Desta maneira é informado o argumento final de número 2, o que significa que a troca só ocorrerá na segunda ocorrência.

COMBINANDO AS FUNÇÕES

As combinações vão variar de acordo com a necessidade do usuário para realizar o tratamento dos dados ou textos, estes exemplos, demonstram as possibilidades de combinações, porém podem ser criadas diversas outras de acordo com os cenários encontrados.

Exemplo:



	A	B	C	D	E
1					
2	Código item	Esquerda	Direita		
3	D2123044.Camisa_Gola_V	D2123044	Camisa_Gola_V		
4	D23551.Calça_Jeans44	D23551	Calça_Jeans44		
5	D31002.Calça_Social42	D31002	Calça_Social42		
6	D310342.Calça_Social40	D310342	Calça_Social40		
7	D31003212.Calça_Social38	D31003212	Calça_Social38		

=ESQUERDA(A3;LOCALIZAR(".",A3;1)-1)

A3 – Célula que será aplicada a função esquerda

"." – Texto que se deseja localizar, neste exemplo deseja-se localizar o ponto.

A3 – Célula que será aplicada a função localizar

1 – A partir de qual caractere será localizado.

- **1** – Numero subtraído do localizar, para se obter a posição do ponto -1 unidade.

O leitor pode-se questionar porque não utilizar “=ESQUERDA(A3;8)”, afinal se

obteria o mesmo resultado de forma mais simples. Porém quando deseja-se aplicar esta função para diversos elementos que possuem caracteres variados como neste exemplo, é preciso variar o número dentro da função esquerda, para tanto se utilizar a função: “LOCALIZAR(“.”;A3;1)-1” neste exemplo no lugar do número 8, para desta maneira, sempre se localizar a quantidade de caracteres até o ponto (“.”), subtraindo uma unidade, para se aplicar esquerda até antes do mesmo. Desta forma pode-se arrastar a função para quantos elementos forem necessários.

Exemplo 2:

	A	B	C	D	E	F
1						
2	Código Item	Esquerda	Direita			
3	D2123044.Camisa_Gola_V	D2123044	Camisa_Gola_V			
4	D23551.Calça_Jeans44	D23551	Calça_Jeans44			
5	D31002.Calça_Social42	D31002	Calça_Social42			
6	D310342.Calça_Social40	D310342	Calça_Social40			
7	D31003212.Calça_Social38	D31003212	Calça_Social38			

=DIREITA(A3;NÚM.CARACT(A3)-LOCALIZAR(“.”;A3;1))

A3 – Célula que será aplicada a função direita

A3 – Célula que será aplicada a função núm.caract

“.” – Texto que se deseja localizar, neste exemplo deseja-se localizar o ponto.

A3 – Célula que será aplicada a função localizar

1 – A partir de qual caractere será localizado.

Com a formula direita, utiliza-se também o elemento NÚM.CARACT(), uma vez que o LOCALIZAR() conta a partir da esquerda, é preciso utilizar a subtração do total de caracteres com a função localizar, para se obter o número de caracteres desejados para aplicar a função DIREITA().

TEXTO()

A função texto é extremamente importante para lidar com dados, os quais muitas

vezes estão em formatos indesejados, seja uma data que está como número, ou para padronizar formatações e etc.

Exemplo:

	A	B	C	D	E
1	10,4	R\$ 10,40			
2	18,58	R\$ 18,58			
3	10,8889	R\$ 10,89			
4	18,4	R\$ 18,40			


`=TEXTO(A1;"R$ ###0,00")`

A1 – Célula que receberá a formatação

"R\$ ###0,00" – Formatação aplicada

A função é extremamente simples, exigindo apenas duas entradas de dado, uma é a célula que será formatada (Neste exemplo **A1**) e a outra é o código da formatação (Neste exemplo o código: **"R\$ ###0,00"**, formato de moeda) . Desta forma, é possível garantir que uma coluna estará sempre com a formatação desejada através da função.

Obs 1: Diversos exemplos de código de formatação podem ser consultados no menu através da opção “Mais Formatos de Números”, Número, Personalizado.



Exemplo:

	A	B	C	D
1	0,242	24,2%		
2	0,234	23,4%		
3	0,08	8,0%		
4	0,154	15,4%		

`=TEXTO(A1;"0,0%")`

A1 – Célula que receberá a formatação

"R\$ ###0,00" – Formatação aplicada

O exemplo é extremamente semelhante ao anterior, porém ao invés de aplicar a formatação de moeda, é aplicada a formatação de percentual (**R\$ ###0,00"**) na célula **A1**.

Dica:

É possível transformar datas em textos de diversas maneiras diferentes, para tanto basta utilizar a fórmula texto e utilizar os códigos de formatação, segue abaixo alguns exemplos de aplicações da função texto para datas.

Data	Texto	Fórmula Utilizada
05/01/2018	05/01/18	=TEXTO(B3;"dd/mm/aa")
05/01/2018	05 jan 2018	=TEXTO(A3;"dd mmm aaa")
05/01/2018	sex 05 jan	=TEXTO(A4;"ddd dd mmm")
05/01/2018	janeiro 2018	=TEXTO(A5;"mmmm aaaa")
05/01/2018	05 janeiro 2018	=TEXTO(A6;"dd mmmm aaaa")

ARRUMAR()

Esta função é essencial para usuários que lidam com muitos dados. É sempre comum lidar com informações que estão desajustadas e possuem espaços no início ou no final da célula que não deveriam ter. Para solucionar este problema, utiliza-se a função ARRUMAR()

Exemplo:

	A	B	C	D	E
1	Antes	Depois			
2	D1906	D1906			
3	D8386	D8386			
4	D3150	D3150			
5	D5418	D5418			
6	D6116	D6116			
7	D5886	D5886			
8	D9087	D9087			
9	D5370	D5370			
10	D6045	D6045			
11	D8905	D8905			

=ARRUMAR(A2)

A2 – Célula onde a função será aplicada

A fórmula é extremamente simples, seu único argumento é a célula desejada para que os espaços de prefixo e sufixo sejam retirados. Desta forma observa-se que após a aplicação da função na coluna A, ela fica sem os problemas de desajustes na coluna B.

LETRAS MAIÚSCULAS

Existem três funções básicas para controlar se as letras serão maiúsculas, minúsculas ou se terão apenas o seu primeiro caractere maiúsculo. Qualquer necessidade além destes três casos, deverá ser solucionada com as combinação das funções de manipulação de texto.

Para os três casos deve se utilizar as seguintes funções:

	A	B	C
1	Exemplo	Fórmula	Retorno
2	planilha	=MAIÚSCULA(A2)	PLANILHA
3	planilha	=PRI.MAIÚSCULA(A3)	Planilha
4	PLANILHA	=MINÚSCULA(A4)	planilha

As funções são comumente utilizadas em extrações de dados, onde os textos extraídos de uma base de dados normalmente estão sem nenhuma formatação.

VERIFICAR CÉLULA

Existem funções que funcionam para verificar qual tipo de conteúdo de uma célula, seus retornos são basicamente verdadeiro ou falso. O único argumento deste conjunto de fórmulas é a célula que se deseja verificar.

	A	B	C
1	Exemplo	Fórmula	Retorno
2	Texto	=ÉTEXTO(A2)	VERDADEIRO
3	10	=ÉNÚM(A3)	VERDADEIRO
4	2	=ÉFÓRMULA(A4)	VERDADEIRO
5		=ÉCÉL.VAZIA(A5)	VERDADEIRO
6	#DIV/0!	=ÉERROS(A6)	VERDADEIRO
7	3	=ÉIMPAR(A7)	VERDADEIRO
8	4	=ÉPAR(A8)	VERDADEIRO
9	FALSO	=ÉLÓGICO(A9)	VERDADEIRO

Essas são as principais funções que verificam o conteúdo de uma célula, no exemplo acima todas foram utilizadas com o propósito de retornar seus valores como verdadeiro.

FUNÇÕES DE LOCALIZAÇÃO

As funções de localização, tem por objetivo retornar, valores, textos ou termos baseados em um valor de busca. Existem diversas funções que executam esse procedimento, as mais conhecidas são PROCV() e a combinação ÍNDICE(CORRESP()), esta última sendo considerada a mais importante, porém para apresentar esta, primeiramente serão apresentados o ÍNDICE() e o CORRESP() individualmente.

ÍNDICE()

Esta função é extremamente simples, através de coordenadas, ela retorna o valor da célula desejada. Este processo funciona em uma direção, ou seja, informando a região de uma linha ou coluna e a posição desejada ou informando uma matriz e duas coordenadas.

Exemplo 1:

	A	B	C	D	E
1	Departamento	Vendas			Índice
2	Feminino	23.887,00			Masculino
3	Masculino	16.992,00			
4	Infantil	89.002,00			
5	Infante-Juvenil	12.887,00			
6	Lingerie	19.009,00			
7	Calçados Masculinos	5.664,00			
8	Calçados Femininos	9.988,00			
9	Calçados Infantis	3.821,00			
10	Acessórios	1.203,00			

=ÍNDICE(A2:A10;2)

B2:B10 – Região que retorna os valores

2 – Linha de retorno do valor

Neste exemplo, utiliza-se a formula índice para encontrar quem está na segunda linha, **B2:B10** é a região, e 2 é a posição da linha retornada. O resultado consequentemente será Masculino.

Exemplo 2:

E2		:	✕ ✓ f _x			=ÍNDICE(A1:C3;1;3)	
	A	B	C	D	E	F	
1	245	13	421		índice		
2	123	64	2312		421		
3	35	53	321				
4							

=ÍNDICE(A1:C3;1;3)

A1:C3 – Região que retorna os valores

1 – Linha onde será retornado o valor

3 – Coluna onde será retornado o valor

Esta fórmula, aplica o índice para uma matriz de região **A1:C3**, como se trata de uma região de matriz, é preciso informar duas coordenadas para retornar o valor, primeiro a linha e depois a coluna, neste exemplo de valor **1** e **3**, respectivamente.

CORRESP()

Esta função funciona de forma inversa ao índice, ao informar um valor e uma região, a posição é retornada, ao contrário do índice, esta funciona apenas em uma direção.

Exemplo:

	A	B	C	D	E	F
1	Departamento	Vendas			Índice	Corresp
2	Feminino	23.887,00			Masculino	2
3	Masculino	16.992,00				
4	Infantil	89.002,00				
5	Infanto-Juvenil	12.887,00				
6	Lingerie	19.009,00				
7	Calçados Masculinos	5.664,00				
8	Calçados Femininos	9.988,00				
9	Calçados Infantis	3.821,00				
10	Acessórios	1.203,00				

=CORRESP("Masculino";A2:A10;0)

"Masculino" – Nome procurado

A2:A10 – Região procurada

0 – Correspondência exata

Desta forma, o CORRESP() retorna a posição da palavra masculino, dentro da seleção, logo o valor será 2, visto que da seleção informada, ele está na segunda posição, o número 0, informado após a região, determina que a correspondência será exata.

Agora que o ÍNDICE() e o CORRESP() já foram apresentados, será possível apresentar a poderosa combinação das duas funções. Uma vez que o índice retorna o valor de uma posição e o CORRESP() retorna a posição de um valor, será simples entender a utilidade desta duas funções juntas.

ÍNDICE(CORRESP())

Esta combinação, tem uma poderosa função de buscar o conteúdo de uma célula em uma região e retornar o valor de uma outra região. Como apresentado anteriormente, o ÍNDICE() tem a função de retornar o valor de uma coordenada, o CORRESP() por sua vez realiza o processo inverso, ao utiliza-los em colunas diferentes, é possível buscar qualquer valor, seja em uma direção ou em duas

direções (apresentado na próxima seção).

Dica: Muitos falam sobre a importância do PROCV(), mas nem todos sabem que a fórmula índice com suas combinações tem um potencial muito maior que o PROCV(), realizando a mesma função, porém com maior eficiência e simplicidade e sem limitações indesejadas.

Exemplo:

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F	G	H	I
1	Departamento	Vendas			Índice + Corresp				
2	Feminino	23.887,00			Feminino	=ÍNDICE(B2:B10;CORRESP(E2;A2:A10;0))			
3	Masculino	16.992,00			Masculino	16.992,00			
4	Infantil	89.002,00			Acessórios	1.203,00			
5	Infanto-Juvenil	12.887,00							
6	Lingerie	19.009,00							
7	Calçados Masculinos	5.664,00							
8	Calçados Femininos	9.988,00							
9	Calçados Infantis	3.821,00							
10	Acessórios	1.203,00							

=ÍNDICE(B2:B10;CORRESP(E2;A2:A10;0))

B2:B10 – Região que retorna os valores

E2 – Valor procurado

A2:A10 – Região procurada

0 – Correspondência exata

Neste exemplo, deseja-se buscar o valor de vendas do departamento “Feminino”, para tanto, informa-se que a região de índice é **B2:B10**, posteriormente, é utilizado o CORRESP() buscando a célula de valor “Feminino”, célula **E2**, na região **A2:A10**. Esta combinação fará o valor correspondente ser retornado na célula onde a função está sendo aplicada.

Dica: Como muitas vezes se deseja arrastar esta fórmula para os demais itens buscados, é interessante “travar as fórmulas” com \$, desta maneira,

pode-se arrastar sem modificar a região de busca, da seguinte forma:

=ÍNDICE(\$B\$2:\$B\$10;CORRESP(E2;\$A\$2:\$A\$10;0))

E	F
Índice + Corresp	
Feminino	23.887,00
Masculino	16.992,00
Acessórios	1.203,00
Lingerie	
Infantil	



E	F
Índice + Corresp	
Feminino	23.887,00
Masculino	16.992,00
Acessórios	1.203,00
Lingerie	19.009,00
Infantil	89.002,00

ÍNDICE(CORRESP()); CORRESP()

Como visto anteriormente, o índice retorna um valor através de uma coordenada, porém só foi apresentado o índice quando aplicado em uma coluna, porém o índice pode ser aplicado em duas direções, através de duas coordenadas. Conforme exemplo abaixo:

Exemplo:

		Vendas					ÍNDICE
2	Departamento	Janeiro	Fevereiro	Março	Abril		18.691,20
3	Feminino	23.887,00	26.275,70	28.903,27	31.793,60		
4	Masculino	16.992,00	18.691,20	20.560,32	22.616,35		
5	Infantil	89.002,00	97.902,20	107.692,42	118.461,66		
6	Infanto-Juvenil	12.887,00	14.175,70	15.593,27	17.152,60		
7	Lingerie	19.009,00	20.909,90	23.000,89	25.300,98		
8	Calçados Masculinos	5.664,00	6.230,40	6.853,44	7.538,78		
9	Calçados Femininos	9.988,00	10.986,80	12.085,48	13.294,03		
10	Calçados Infantis	3.821,00	4.203,10	4.623,41	5.085,75		
11	Acessórios	1.203,00	1.323,30	1.455,63	1.601,19		

Neste exemplo, através do índice selecionando a região de todos os valores, B3:E11, deseja-se saber o valor da posição 2;2 ou seja, linha 2 da região e coluna 2. Conseqüentemente o valor retornado é equivalente a Masculino para o mês de Fevereiro.

Da mesma forma como apresentado anteriormente, os valores do ÍNDICE() serão representados pela função CORRESP(), porém desta vez, será realizado de forma dupla, para retornar o departamento (linha), e o mês desejado (coluna).

Exemplo:

		Vendas				Índice + Corresp + Corresp	
Departamento	Janeiro	Fevereiro	Março	Abril			
Feminino	23.887,00	26.275,70	28.903,27	31.793,60	Março	Abril	
Masculino	16.992,00	18.691,70	20.560,32	22.616,35	infantil	=ÍNDICE(B3:E11;CORRESP(G4;A3:A11;0);CORRESP(H3;B2:E2;0))	
Infantil	89.002,00	97.902,20	107.692,42	118.461,66	Lingerie		
Infante-Juvenil	12.887,00	14.175,70	15.591,27	17.152,60	Acessórios		
Lingerie	19.009,00	20.909,90	23.000,89	25.300,98			
Calçados Masculinos	5.664,00	6.230,40	6.853,44	7.538,78			
Calçados Femininos	9.988,00	10.988,80	12.085,48	13.294,03			
Calçados Infantis	3.821,00	4.203,10	4.623,41	5.085,75			
Acessórios	1.203,00	1.323,30	1.458,63	1.601,18			

=ÍNDICE(B3:E11;CORRESP(G4;A3:A11;0);CORRESP(H3;B2:E2;0))

B3:B11 – Região que retorna os valores

G4 – Valor procurado 1

A3:A11 – Região procurada 1

0 – Correspondência exata

H3 – Valor procurado 2

B2:E2 – Região procurada 2

0 – Correspondência exata

O funcionamento é extremamente o mesmo, porém desta vez se busca duas correspondências, no caso deste exemplo, uma para a célula **G4**, relacionada aos departamentos, com região de procura **A3:A11** nos departamentos e a célula **H3**, relacionada aos meses do ano, na região de procura em **B2:E2**, de maneira semelhante, a fórmula do CORRESP fecha com o valor 0, para determinar que a correspondência que se busca é exata.

Dica: Da mesma maneira como apresentado anteriormente, é interessante travar as regiões de retorno e de busca, para que desta forma, se possa arrastar a fórmula para demais variáveis buscadas, conforme exemplo abaixo:

	A	B	C	D	E	F	G	H	I
1		Vendas						Índice + Corresp + Corresp	
2	Departamento	Janeiro	Fevereiro	Março	Abril				
3	Feminino	23.887,00	26.275,70	28.903,27	31.793,60			Março	Abril
4	Masculino	16.992,00	18.691,20	20.560,32	22.616,35		Infantil	107.692,42	118.461,66
5	Infantil	89.002,00	97.902,20	107.692,42	118.461,66		Lingerie	23.000,89	25.300,98
6	Infanto-Juvenil	12.887,00	14.175,70	15.593,27	17.152,60		Acessórios	1.455,63	1.601,19
7	Lingerie	19.009,00	20.909,90	23.000,89	25.300,98				
8	Calçados Masculinos	5.664,00	6.230,40	6.853,44	7.538,78				
9	Calçados Femininos	9.988,00	10.986,80	12.085,48	13.294,03				
10	Calçados Infantis	3.821,00	4.203,10	4.623,41	5.085,75				
11	Acessórios	1.203,00	1.323,30	1.455,63	1.601,19				

=ÍNDICE(\$B\$3:\$E\$11;CORRESP(\$G4; \$A\$3:\$A\$11;0);CORRESP(H\$3; \$B\$2:\$E\$2;0))

Obs1: \$G4, recebeu a trava apenas na coluna, pois ainda interessa arrasta-lo para baixo.

Obs2: H\$3, recebeu a trava apenas na linha, pois ainda interessa arrasta-lo para os lados.

PROCV()

Esta é uma das funções mais conhecidas para se buscar um valor no Excel, através dela, é possível informar um valor procurado, uma matriz de busca e obter-se um valor correspondente, para facilitar a comparação, será utilizado o mesmo exemplo utilizado para o ÍNDICE(CORRESP()).

Exemplo:

	A	B	C	D	E	F	G
1	Departamento	Vendas			ProcV		
2	Feminino	23.887,00		Feminino	=PROCV(D2;A2:B10;2;FALSO)		
3	Masculino	16.992,00		Masculino	16.992,00		
4	Infantil	89.002,00		Acessórios	1.202,00		
5	Infanto-Juvenil	12.887,00					
6	Lingirie	19.009,00					
7	Calçados Mascu	5.664,00					
8	Calçados Femin	9.988,00					
9	Calçados Infant	3.821,00					
10	Acessórios	1.202,00					

=PROCV(D2;A2:B10;2;FALSO)

D2 – Célula que se deseja buscar

A2:B10 – Matriz de procura

2 – Posição da coluna de retorno

FALSO – Busca exata pelo termo procurado

A função tem por objetivo, buscar um valor, neste exemplo célula **D2**, dentro da região **A2:B10**, retornando o valor equivalente da coluna **2** da matriz informada. Utiliza-se **FALSO** para especificar que o termo buscado terá correspondência exata e não aproximada.

Obs: O PROCV() possui uma limitação, onde o valor de retorno precisa obrigatoriamente estar a direita do valor procurado, funcionando apenas para estruturas organizadas da esquerda para direita. Caso contrário será necessário alterar a forma como os dados estão dispostos na planilha ou utilizar a combinação ÍNDICE(CORRESP()) que tem o mesmo

funcionamento, sem esta limitação.

Dica: Como muitas vezes se deseja arrastar esta fórmula para os demais itens buscados, é interessante “travar as fórmulas” com \$, desta maneira, pode-se arrastar sem modificar a região de busca, da seguinte forma:

= PROCV(D2;\$A\$2:\$B\$10;2;FALSO)

PROCH()

Esta função funciona de forma extremamente similar ao PROCV(), a única diferença é que o PROCH() realiza uma busca de forma horizontal ao invés de ser na vertical. A fórmula é eficaz para realizar pesquisa e retorno de dados.

Exemplo:

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2												
3	Dia		1	2	3	4	5	6	7	8	9	10
4	Valor		61,0	176,0	99,0	134,0	166,0	188,0	101,0	148,0	159,0	196,0
5												
6												
7		Dia Procurado	Valor									
8		5	166,0									
9		7	101,0									
10		9	159,0									

=PROCH(B8;C3:L4;2;FALSO)

B8 - Valor procurado

C3:L4 - Matriz onde será o valor será encontrado

2 - Será buscado na segunda linha da matriz

FALSO - A correspondência será exata

Este exemplo demonstra de forma direta a aplicação do PROCH(), onde é necessário pesquisar um valor que está presente na célula **B8**, dentro da matriz **C3:L4**, para tanto se utiliza o índice de valor **2**, o que significa que o valor será encontrado na segunda linha desta matriz. O último argumento é dado como **FALSO**, pois se deseja buscar o exato valor e não o seu número aproximado.

Obs: É sempre interessante "travar" a região de procura para poder arrastar a fórmulas para os demais valores procurados, para tanto utiliza-se \$, a função fica da seguinte forma: =PROCH(**B8**;**\$C\$3:\$L\$4**;2;FALSO)

PROCV(MAIOR())

Esta função tem por objetivo realizar uma busca e retornar o valor correspondente a uma célula de maior valor.

Dica: É importante ressaltar que a mesma combinação também pode ser utilizada para PROCV(MENOR())

Exemplo:

	A	B	C	D	E	F	G	H
1	Nota	Aluno						
2	33	Pedro						
3	30	João						
4	80	Diogo						
5	48	José						
6	58	Ana						
7	98	Clara			Maior Nota	Clara		
8	85	Maria						
9	96	Carlos						
10	46	Alice						
11								

=PROCV(MAIOR(A2:A10;1);A2:B10;2;FALSO)

A2:A10 – Região de máximo valor desejado

1 – Valor buscado é o maior (Para segundo maior, usar 2 e assim sucessivamente)

A2:B10 – Matriz de busca

2 – Da matriz selecionada o dado será retornado da segunda coluna

FALSO – Utilizado para valor exato

A função funciona da maneira padrão do PROCV(), porém ao invés de se definir um valor de busca, utiliza-se MAIOR(A2:A10;1). A Matriz de procura é a região onde a informação será pesquisada A2:B10, e o índice 2, indica que o valor retornado estará na segunda coluna da matriz de procura.

DESLOC()

Esta é mais uma das funções que muitos pensam não ter utilidade, porém ela é extremamente poderosa, principalmente quando combinada com outras funções. Seu funcionamento basicamente consiste em retornar o valor de células a uma distância informada a partir de uma referência inicial.

Exemplo:

D2		:	X	✓	<i>fx</i>	=DESLOC(B2;5;2)
	A	B	C	D	E	F
1						
2		Referencia		Retorno		
3						
4						
5	5 Linhas					
6						
7				Retorno		
8		2 Colunas				

=DESLOC(B2;5;2)

B2 – Referência de início

5 – Quantidade de linhas deslocadas

2 – Quantidade de colunas deslocadas

Uma vez que a fórmula seja aplicada na célula **B2** com deslocamento de 5 linhas e 2 colunas o valor encontrado é exatamente o da célula D7 que possui texto Retorno.

Obs: As setas são apenas ilustrativas neste exemplo.

DESLOC(CORRESP())

Esta é mais uma combinação que tem alto potencial para realizar busca de informações, assim como o ÍNDICE(CORRESP()), esta também é uma

alternativa poderosa para o ProcV, possuindo também menores limitações e funcionando de forma mais eficaz.

Exemplo:

SOMA =DESLOC(B3;CORRESP(H3;B4:B13;0);1)

	A	B	C	D	E	F	G	H	I
1									
2		Loja de Calçados						Relatório de Venda de Vestuário	
3		Departamento	Venda					Vestuário Esportivo Masc	11.235,46
4		Tênis Masculino	32.553,65					Vestuário Esportivo Fem	16.431,10
5		Tênis Feminino	21.356,43					Vestuário Esportivo Inf	8.765,21
6		Tênis Infanto-Juvenil	12.345,31						
7		Tênis Infantil	1.245,21					Relatório de Venda de Tênis	
8		Meias	2.345,43					Tênis Masculino	32.553,65
9		Material Esportivo	3.135,23					Tênis Feminino	21.356,43
10		Acessórios	12.356,56					Tênis Infanto-Juvenil	12.345,31
11		Vestuário Esportivo Masc	11.235,46					Tênis Infantil	1.245,21
12		Vestuário Esportivo Fem	16.431,10						
13		Vestuário Esportivo Inf	8.765,21					Relatório de Vendas Diversas	
14								Meias	2.345,43
15								Material Esportivo	3.135,23
16								Acessórios	12.356,56

=DESLOC(B3;CORRESP(H3;B4:B13;0);1)

B3 – Célula referência para a função realizar o deslocamento

H3 – Célula que será buscada na região de procura

B4:B13 – Região de procura

0 – Correspondência exata

1 – A partir da célula referência (B3), será deslocado 1 coluna para direita

O funcionamento desta combinação se dá através da função DESLOC() que retorna um valor a partir de uma referência, a referência neste exemplo é a célula **B3** a partir da qual precisa de “coordenadas” para encontrar o valor desejado, algo que a função CORRESP() realiza, através da célula de procura **H3**, na região de busca **B4:B13**, desta forma o valor da linha desejada é encontrada, a da coluna é basicamente o número **1**, visto que a coluna dos valores fica exatamente

a uma unidade para direita.

Obs: É sempre interessante “travar” as fórmulas para poder arrasta-las sem deslocar as referências corretas, no caso deste exemplo:
=DESLOC(B\$3;CORRESP(H3;\$B\$4:\$B\$13;0);1)

INDIRETO()

A função indireto é extremamente poderosa dentro do Excel. Ela utiliza referencia textual para retornar o valor de outra célula, seja ela na mesma aba ou não. Ela tem o potencial de substituir a utilização de diversas fórmulas condicionais, simplificando a consolidação de dados. O primeiro exemplo tem por objetivo apenas demonstrar a utilização de forma simples, posteriormente outro exemplo será apresentado demonstrando sua real utilidade.

Exemplo 1:

	A	B	C	D	E
1					
2	4	8		1	
3	D5	D9		2	
4				3	
5				4	
6				5	
7				6	
8				7	
9				8	

=INDIRETO(A3;VERDADEIRO)

A3 = Referência para retornar o valor

VERDADEIRO = Referência no estilo "A1", Falso retorna "L1C1"

Os argumentos da função são extremamente simples, basicamente **A3** é célula que será utilizada como referência para retornar um valor. Como a célula **A3** possui seu valor D5, o valor encontrado pela função indireto é o mesmo da célula D5. O segundo argumento é opcional, não preencher é o mesmo que optar por Verdadeiro, o que especifica a referência do tipo "A1", para alternar para referência do tipo L1C1 é preciso colocar como Falso.

CONT.VALORES()

Esta função é simples, tem o objetivo de retornar o valor de células preenchidas em uma região informada. Ela possui uma excelente funcionalidade para ser combinada com o índice, a qual será apresentada na seção a seguir.

Exemplo:

	A	B	C	D	E	F	G	H	I	J	K	L	M
1		Vendas											
2	Departamento	Janeiro	Fevereiro	Março	Abril	Mai	Junho	Julho	Agosto	Setembro	Outubro	Novembro	Dezembro
3	Feminino	23.887,00	26.275,70	28.901,27	31.791,00								
4	Masculino	16.992,00	18.091,20	20.560,22	22.816,25								
5	Infantil	89.002,00	97.902,20	107.892,42	118.461,66								
6	Infanto-juvenil	12.887,00	14.125,70	15.391,27	17.122,60								
7	Lingerie	19.009,00	20.909,90	21.000,89	23.300,98								
8	Calçados Masculinos	5.694,00	6.230,40	6.831,44	7.316,70								
9	Calçados Femininos	9.988,00	10.986,80	12.085,48	13.294,03								
10	Calçados Infantis	3.822,00	4.201,20	4.622,41	5.085,75								
11	Accessórios	1.203,00	1.323,30	1.455,63	1.601,29								
12													
13													
14													
15	Valores Atuais												
16	Feminino	=CONT.VALOR											
17	Masculino												
18	Infantil												
19	Infanto-juvenil												
20	Lingerie												
21	Calçados Masculinos												
22	Calçados Femininos												
23	Calçados Infantis												
24	Accessórios												

15	Valores Atuais	
16	Feminino	4

=CONT.VALORES(**B3:M3**)

B3:M3 – Região a ser contada

Informada a região **B3:M3**, o valor retornado será a quantidade de valores

preenchidos na linha especificada.

Exemplo 2:

C2 fx =INDIRETO(B2&C1)

	A	B	C	D	E	F	G	H	I
1			IA1	IA2	IA3				
2		Janeiro	22	36	92				
3		Fevereiro	44	19	12				
4		Março	91	36	76				
5		Abril	48	9	62				
6		Mai	19	80	64				
7		Junho	37	17	19				
8		Julho	33	74	13				
9		Agosto	82	59	77				
10		Setembro	70	68	40				
11		Outubro	27	64	18				
12		Novembro	52	11	59				
13		Dezembro	10	39	72				
14									

Principal Janeiro Fevereiro Março Abril Maio Junho Julho Agosto

P27 fx

	A	B	C	D
1	22			
2	36			
3	92			
4				
5				

Principal Janeiro Fevereiro

=INDIRETO(B2&C1)

B2 – Referência para busca

& - Concatena os dois valores

C1 - Referência para busca

Este exemplo já demonstra um potencial maior da função indireto, através da mesma, pode-se consolidar dados de outras abas, utilizando a fórmula indireto com o nome da aba concatenado com a célula desejada, no caso da célula selecionada indireto está buscando o valor da célula **B2** (Janeiro), concatenado com **C1** (!A1), o valor buscado de maneira indireta será Janeiro!A1.

Obs: É sempre interessante "travar" a linha da região horizontal e a coluna da região vertical, para que as mesmas tenham suas posições mantidas quando as fórmulas forem arrastadas. Neste exemplo ficaria:
=INDIRETO(**B2&C1**)

INDICE(CONT.VALORES())

Esta combinação é muito útil quando se deseja retornar sempre o último valor disponível em uma região desejada.

Exemplo:

The screenshot shows an Excel spreadsheet with a table of sales data. The table has columns for months (January to December) and rows for different departments and categories. The formula bar shows the formula =INDICE(B3:M3;CONT.VALORES(B3:M3)). The result of the formula is 31.793,60.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1		Vendas											
2	Departamento	Janeiro	Fevereiro	Março	Abril	Maió	Junho	Julho	Agosto	Setembro	Outubro	Novembro	Dezembro
3	Feminino	23.887,00	26.275,70	28.903,27	31.793,60								
4	Masculino	16.992,00	18.691,20	20.580,32	22.616,35								
5	Infantil	89.002,00	97.902,20	107.692,42	118.482,66								
6	Infanto-Juvenil	12.887,00	14.175,70	15.593,27	17.152,60								
7	lingerie	19.008,00	20.908,20	23.000,88	25.300,96								
8	Calçados Masculinos	5.694,00	6.230,40	6.853,44	7.516,78								
9	Calçados Femininos	9.988,00	10.988,80	12.083,48	13.294,03								
10	Calçados infantis	3.823,00	4.203,30	4.623,41	5.085,75								
11	Acessórios	1.203,00	1.323,30	1.455,63	1.601,19								
12													
13													
14													
15	Valores Atuais												
16	Feminino												31.793,60

=ÍNDICE(**B3:M3**;CONT.VALORES(**B3:M3**))

B3:M3 – Região de retorno

B3:M3 – Região de contagem

Sempre que novos valores forem inseridos, a combinação atualizará os valores para os últimos inseridos, conforme na imagem abaixo, onde demonstra que uma vez que os valores de maio sejam preenchidos, a função atualizará automaticamente os valores.

Exemplo:

	A	B	C	D	E	F	G	H	I	J	K	L	M
	Vendas												
2	Departamento	Janeiro	Fevereiro	Março	Abril	Maio	Junho	Julho	Agosto	Setembro	Outubro	Novembro	Dezembro
3	Feminino	23.887,00	26.275,70	28.903,27	31.793,00	31.504,56							
4	Masculino	16.992,00	18.691,20	20.560,32	22.616,35	22.410,75							
5	Infantil	89.062,00	97.902,20	107.692,42	118.461,66	117.384,74							
6	Infanto-Juvenil	12.887,00	14.175,70	15.593,27	17.132,00	16.996,66							
7	Lingerie	19.095,00	20.969,90	23.000,89	25.300,98	25.070,97							
8	Calçados Masculinos	5.464,00	6.230,40	6.853,44	7.538,78	7.470,2496							
9	Calçados Femininos	9.988,00	10.986,80	12.085,40	13.294,03	13.173,1732							
10	Calçados Infantis	1.821,00	4.203,10	4.623,41	5.085,75	5.039,5169							
11	Acessórios	1.263,00	1.323,30	1.455,63	1.601,19	1.586,4367							
12													
13													
14													
15	Valores Atuais												
16	Feminino	31.504,56											
17	Masculino	22.410,75											
18	Infantil	117.384,74											
19	Infanto-Juvenil	16.996,66											
20	Lingerie	25.070,97											
21	Calçados Masculinos	7.470,25											
22	Calçados Femininos	13.173,17											
23	Calçados Infantis	5.039,52											
24	Acessórios	1.586,54											

SEERRO() + FUNÇÕES DE PROCURA

A função SEERRO() é comumente utilizada acompanhando funções de procura de dados, como PROCV(), ÍNDICE(CRRESP()) ou DESLOC(CORRESP()). Uma vez que é esperado que nem todos os dados sejam encontrados, desta forma, todos os que não forem encontrados apresentarão um erro de #N/D, conforme o exemplo a seguir.

Exemplo:

Formula bar: `=ÍNDICE(C4:C10;CORRESP(F6;B4:B10;0))`

	A	B	C	D	E	F	G	H
1								
2		Loja de Calçados				Relatório de Venda de Vestuário		
3		Produto	Venda			Smartphone	32.553,65	
4		Smartphone	32.553,65			Tablet	2.345,43	
5		Notebook	21.356,43			Video Game	3.135,23	
6		Desktop	12.345,31			Ventilador	#N/D	
7		Televisão	1.245,21					
8		Tablet	2.345,43					
9		Video Game	3.135,23					
10		Conversor	12.356,56					
11								
12								

Supondo que esta busca já seja esperada e deseja-se que ela apenas retorne o valor zero, basta utilizar a função dentro de um SEERRO() com argumento de valor zero.

Formula bar: `=SEERRO(ÍNDICE(C4:C10;CORRESP(F6;B4:B10;0));0)`

	A	B	C	D	E	F	G	H
1								
2		Loja de Calçados				Relatório de Venda de Vestuário		
3		Produto	Venda			Smartphone	32.553,65	
4		Smartphone	32.553,65			Tablet	2.345,43	
5		Notebook	21.356,43			Video Game	3.135,23	
6		Desktop	12.345,31			Ventilador	-	
7		Televisão	1.245,21					
8		Tablet	2.345,43					
9		Video Game	3.135,23					
10		Conversor	12.356,56					

Desta forma, o inconveniente #N/D é substituído pelo valor zero.

FUNÇÕES DE SOMA E MÉDIA

Existem diversas formas de somar informações no Excel, as quais vão desde fórmulas mais simples até as mais complexas, permitindo ao usuário somar dados de maneira direta ou somar com a utilização de uma condição. A seguir os diversos casos serão apresentados e exemplificados.

SOMA()

Esta é uma das funções mais básicas do Excel, consiste basicamente em escolher a região que será somada.

Exemplo:

	A	B
1		31
2		82
3		53
4		20
5		45
6		87
7	Total	=SOMA(B1:B6)

	A	B
1		31
2		82
3		53
4		20
5		45
6		87
7	Total	318

=SOMA(B1:B6)

B1:B6 – Intervalo onde a soma será realizada

A função é simples e direta, ela basicamente insere a soma de uma região de valores escolhidos.

SOMARPRODUTO ()

Esta função tem por objetivo somar o produto de duas regiões, ela é extremamente simples e substitui a utilização de auxiliares para fazer o mesmo procedimento utilizando apenas Soma().

Exemplo:

	A	B	C	D	E
1	12	57	=SOMARPRODUTO(B1:B7;A1:A7)		
2	74	13			
3	77	60			
4	59	85			
5	73	14			
6	36	62			
7	29	87			



	A	B	C
1	81	70	18201
2	86	57	
3	39	73	
4	66	24	
5	46	17	
6	16	34	
7	26	72	

=SOMARPRODUTO(B1:B7;A1:A7)

B1:B7 – Primeira Matriz

A1:A7 – Segunda Matriz

Neste exemplo, são selecionadas duas regiões que serão multiplicadas entre si e posteriormente somadas posição por posição para chegar a um resultado final.

Obs 1: Neste exemplo foram utilizadas apenas duas regiões, porém a função permite a escolha do número desejado de regiões que serão multiplicadas. É necessário que elas possuam a mesma quantidade de dados.

SUBTOTAL()

Uma característica da função soma apresentada anteriormente, é que a mesma sempre irá somar os valores de uma região, independente das células estarem filtradas ou não. Para contornar este problema, pode-se utilizar a função Subtotal(), a qual pode ser utilizada para realizar a soma apenas nas células desejadas.

Exemplo:

fx			=SUBTOTAL(9;E2:E9)
	D	E	F
	Filtro	405	
	A	27	
	B	53	
	C	9	
	A	96	
	B	34	
	A	72	
	B	35	
	A	79	



fx			=SUBTOTAL(9;E2:E9)
	D	E	F
	Filtro	79	
	A	27	
	A	96	
	A	72	
	A	79	

=SUBTOTAL(9;E2:E9)

9 – Opção escolhida para o subtotal realizar Soma

E2:E9 – Intervalo onde a soma será realizada

Neste exemplo, o subtotal é demonstrado em uma região onde um filtro é aplicado na esquerda dos valores, uma vez que um critério seja estabelecido, a função subtotal automaticamente soma apenas as informações “visíveis”.

Dica: Apesar da função subtotal está sendo mostrada dentro da seção de soma, ela possui as mais variadas utilidades, no exemplo apresentado a função escolhida foi a número 9 que representa a soma, porém existem diversas outras funções para se aplicar o Subtotal(). Segue abaixo uma lista:

Código da função		Função
Inclui ocultos	Ignora ocultos	
1	101	MÉDIA
2	102	CONTA
3	103	CONT.VALORES
4	104	MÁXIMO
5	105	MÍNIMO
6	106	MULT
7	107	DESVPAD
8	108	DESVPADP
9	109	SOMA
10	110	VAR
11	111	VARP

SOMASE()

Esta é uma função de extrema importância para tratamento de dados, através dela é possível somar números seguindo um critério estabelecido.

Exemplo:

	A	B	C	D	E	F	G
1				>15000			
2	A	8.733		37.135			
3	B	9.577					
4	C	5.925					
5	A	5.116					
6	C	19.501					
7	A	162					
8	C	8.477					
9	E	1.920					
10	E	17.634					
11	A	2.050					

=SOMASE(B2:B11;">15000"; B2:B11)

B2:B11 – Intervalo onde é aplicado o critério

">15000" – Critério aplicado

B2:B11 – Intervalo onde a soma será realizada

Este primeiro exemplo, demonstra a aplicação mais simples do SOMASE(). Todo valor dentro do intervalo **B2:B11** que atende o critério de ser maior que 15.000 (">15000"), é somado na célula D2.

Obs 1: Nos casos onde o intervalo de soma é o mesmo de critério, a terceira parte da função não é obrigatória, desta forma, colocar de maneira reduzida, também obtém o mesmo resultado: =SOMASE(**B2:B11**;">15000")

Exemplo 2:

	A	B	C	D	E	F	G
1							
2	A	8.733		A	16.061		
3	B	9.577		E			
4	C	5.925		C			
5	A	5.116					
6	C	19.501					
7	A	162					
8	C	8.477					
9	E	1.920					
10	E	17.634					
11	A	2.050					

=SOMASE(A2:A11;D2;B2:B11)

A2:A11 – Intervalo onde é aplicado o critério

D2 – Critério aplicado

B2:B11 – Intervalo onde a soma será realizada

Neste exemplo, o SOMASE() irá somar todos os valores do intervalo de soma (**B2:B11**) que tiverem valores do intervalo de critério (**A2:A11**) iguais a célula de critério **D2**, ou seja, valores correspondentes a letra A.

	D	E	F	G	H
A		13.848,57			
E		19.554,74			
C		25.425,63			

É interessante travar as posições do intervalo de soma e do intervalo de critério para poder “arrastar” a fórmula para os demais critérios.

=SOMASE(\$A\$2:\$A\$11;D2;\$B\$2:\$B\$11)

SOMASES()

O SOMASES() consiste na mesma lógica do SOMASE(), porém se aplicam mais critérios para soma. Também é trocada a ordem dos intervalos, onde é primeiro solicitado o intervalo de soma

Exemplo:

	A	B	C	D	E	F	G	H
1								
2	V	A	8.733		A	14.010		
3	V	B	9.577		E			
4	F	C	5.925		C			
5	V	A	5.116					
6	F	C	19.501					
7	V	A	162					
8	V	C	8.477					
9	F	E	1.920					
10	V	E	17.634					
11	F	A	2.050					

=SOMASES(C2:C11;B2:B11;E2;A2:A11;"V")

C2:C11 – Intervalo onde a soma será realizada

B2:B11 – Intervalo de critério 1

E2 – Critério 1

A2:A11 – Intervalo de critério 2

"V" – Critério 2

Neste exemplo, o intervalo **C2:C11**, posteriormente são solicitados os critérios, sendo o primeiro **B2:B11**, com critério aplicado a célula **E2**, o segundo **A2:A11** com critério **"V"**.

Obs 1: Caso o intervalo de critério 2 e o critério 2 não fossem adicionados, a função se comportaria como um SOMASE() normal.

fx					
=SOMASES(\$C\$2:\$C\$11;\$B\$2:\$B\$11;E2;\$A\$2:\$A\$11;"V")					
E	F	G	H	I	
A	14.010				
E	17.634				
C	8.477				

Obs 2: É interessante travar as posições do intervalo de soma e do intervalo de critério para poder “arrastar” a formula para os demais critérios.

=SOMASES(\$C\$2:\$C\$11; \$B\$2:\$B\$11;E2;\$A\$2:\$A\$11;"V")

MEDIA()

A função média, consiste em selecionar uma região desejada, então o Excel realiza o cálculo do valor médio relacionado a seleção.

Exemplo:

SOMA					
fx					
=MÉDIA(A1:A7)					
	A	B	C	D	E
1	47		Média	=MÉDIA(A	
2	23				
3	21				
4	81				
5	62				
6	27				
7	64				



D1					
fx					
=MÉDIA(A1:A7)					
	A	B	C	D	E
1	47		Média	46,4286	
2	23				
3	21				
4	81				
5	62				
6	27				
7	64				

=MÉDIA(A1:A7)

A1:A7 – Intervalo onde a média será realizada

Assim como a função SOMA(), a aplicação é direta, consiste em escolher um intervalor onde a média será calculada.

MEDIASE()

O lógica desta função é a mesma da aplicada ao SOMASE(), porém o resultado ao invés de ser uma soma através de um critério, será uma média realizada através de um critério aplicado.

Exemplo:

	A	B	C	D	E	F	G
1							
2	A	8733		A	4015,25		
3	B	9577		E			
4	C	5925		C			
5	A	5116					
6	C	19501					
7	A	162					
8	C	8477					
9	E	1920					
10	E	17634					
11	A	2050					

=MEDIASE(A2:A11;D2;B2:B11)

A2:A11 – Intervalo onde é aplicado o critério

D2 – Critério aplicado

B2:B11 – Intervalo onde a média será realizada

Neste exemplo, o MEDIASE() irá realizar a media todos os valores do intervalo (**B2:B11**) que tiverem valores do intervalo de critério (**A2:A11**) iguais a célula de critério **D2**, ou seja, valores correspondentes a letra A.

	D	E	F	G	H
A		4015,25			
E		9777			
C		11301			

Obs: É interessante travar as posições do intervalo de soma e do intervalo de critério para poder “arrastar” a formula para os demais critérios.

OUTRAS FUNÇÕES MATEMÁTICAS

As funções de soma e média são sem dúvidas as mais utilizadas das funções matemáticas dentro do Excel, porém a plataforma oferece outras funções interessantes e úteis, as quais serão apresentadas nesta seção. Para as quatro primeiras funções apresentadas nesta seção, o mesmo exemplo abaixo será utilizado.

Exemplo:

	A	B
1	Data	Vendas
2	01/01/2018	74
3	02/01/2018	65
4	03/01/2018	98
5	04/01/2018	90
6	05/01/2018	94
7	06/01/2018	59
8	07/01/2018	58
9	08/01/2018	78
10	09/01/2018	76

MÍNIMO ()

Função utilizada para retornar o menor valor de uma seleção de números.

=MÍNIMO(B2:B10)

B2:B10 – Região de procura

Esta função é extremamente simples, utiliza-se a região de procura **B2:B10** onde deseja-se retornar o menor valor, o qual neste exemplo será 58.

MÁXIMO ()

Função utilizada para retornar o maior valor de uma seleção de números.

=MÁXIMO(B2:B10)

B2:B10 – Região de procura

Esta função é extremamente simples, utiliza-se a região de procura **B2:B10** onde deseja-se retornar o maior valor, o qual neste exemplo será 98.

MENOR ()

Função utilizada para retornar o menor valor de uma seleção de números de acordo com a posição informada, por exemplo: ao informar posição 1, a função encontra o menor valor, ao informar a posição 2, a função retorna o segundo menor valor e assim sucessivamente.

=MENOR(**B2:B10**;2)

B2:B10 – Região de procura

2 – Segundo menor valor será retornado

Utilizando-se a região de procura **B2:B10**, é desejado retornar o segundo menor valor, o qual neste exemplo será 59.

Obs: Utilizando o valor 1 no segundo argumento, a função funciona exatamente igual a função MÍNIMO()

MAIOR()

Função utilizada para retornar o maior valor de uma seleção de números de acordo com a posição informada, por exemplo: ao informar posição 1, a função encontra o maior valor, ao informar a posição 2, a função retorna o segundo maior valor e assim sucessivamente.

=MAIOR(**B2:B10**;3)

B2:B10 – Região de procura

3 – Terceiro maior valor será retornado

Utilizando-se a região de procura **B2:B10**, é desejado retornar o terceiro maior valor, o qual neste exemplo será 90.

Obs: Utilizando o valor 1 no segundo argumento, a função funciona exatamente igual a função MÁXIMO()

FATORIAL()

Calcula o valor fatorial de um número.

Exemplo:

SOMA		:	X	✓	f _x	=FATORIAL(A5)
	A	B	C	D	E	
1	1	1				
2	2	2				
3	3	6				
4	4	24				
5	5	120				
6						

=FATORIAL(A5)

A5 – Valor para calcular fatorial

Esta função é simples e direta, basicamente se escolhe uma célula para retornar seu valor em fatorial.

ARREDONADAMENTOS

Existem três funções para realizar o arredondamento, a primeira a ser apresentada basicamente utiliza critérios matemáticos comuns para realizar o arredondamento correto. As outras duas são arredondamentos desejados para cima ou para baixo.

ARRED()

Esta função é utilizada para arredondar os números de acordo com uma quantidade de casas decimais desejadas e seguindo critérios matemáticos.

Exemplo:

SOMA : X ✓ f_x =ARRED(A2;1)

	A	B	C	D	E	F
1	Original	Arred				
2	9,93	9,90				
3	39,73	39,70				
4	42,89	42,90				
5	21,07	21,10				
6	27,07	27,10				
7	4,09	4,10				
8	9,62	9,60				
9	33,46	33,50				

=ARRED(A2;1)

A2 – Célula que será arredondada

1 – Quantidade de casas decimais desejadas

A função realizará o arredondamento de acordo as casas decimais desejadas, desta forma, ao informar o número um, significa que após a aplicação da função o valor retornado será o da célula **A2**, porém com apenas **1** casas decimal. Caso fosse informado o argumento 0, o valor de retorno seria 10, a célula A2.

ARREDONDAR.PARA.BAIXO()

Esta função é utilizada para arredondar os números de acordo com uma quantidade de casas decimais desejadas e seguindo critério de arredondar o valor para baixo.

Exemplo:

SOMA : X ✓ fx =ARREDONDAR.PARA.BAIXO(A2;1)

	A	B	C	D	E	F	G
1	Original	Arred					
2	9,93	9,90					
3	39,73	39,70					
4	42,89	42,80					
5	21,07	21,00					
6	27,07	27,00					
7	4,09	4,00					
8	9,62	9,60					
9	33,46	33,40					

=ARREDONDAR.PARA.BAIXO(A2;1)

A2 – Célula que será arredondada

1 – Quantidade de casas decimais desejadas

A função realizará o arredondamento para baixo de acordo as casas decimais desejadas, desta forma, ao informar o número um, significa que após a aplicação da função o valor retornado será o da célula **A2**, porém com apenas **1** casas decimal.

ARREDONDAR.PARA.CIMA()

Esta função é utilizada para arredondar os números de acordo com uma quantidade de casas decimais desejadas e seguindo critério de arredondar o valor para cima.

Exemplo:

	A	B	C	D	E	F	G
1	Original	Arred					
2	9,93	10,00					
3	39,73	39,80					
4	42,89	42,90					
5	21,07	21,10					
6	27,07	27,10					
7	4,09	4,10					
8	9,62	9,70					
9	33,46	33,50					

=ARREDONDAR.PARA.CIMA(A2;1)

A2 – Célula que será arredondada

1 – Quantidade de casas decimais desejadas

A função realizará o arredondamento para cima de acordo as casas decimais desejadas, desta forma, ao informar o número um, significa que após a aplicação da função o valor retornado será o da célula **A2**, porém com apenas **1** casas decimal.

NÚMEROS ALEATÓRIOS

Números aleatórios são algoritmos complexos dentro da programação, gerar valores com completa distinção na frequência de aparições definitivamente não é uma tarefa simples, o Excel por sua vez, oferece de maneira extremamente simples a geração de valores aleatórios, para tanto basta utilizar as funções que serão apresentadas a seguir.

ALEATÓRIO()

Esta função basicamente tem por objetivo retornar valores aleatórios entre 0 e 1.

Ela pode ser utilizada em combinação com outras funções, como um multiplicador, ou valor de soma, a sua utilidade depende de cada caso.

Exemplo:

A1						
	A	B	C	D	E	F
1	0,9451					
2	0,78226					
3	0,20315					
4	0,41113					
5	0,48379					
6	0,90962					
7	0,28477					

=ALEATÓRIO()

Esta função não possui argumentos, sua aplicação é direta.

ALEATÓRIOENTRE()

Esta função também tem por objetivo gerar números aleatórios, porém como seu próprio nome diz, ela gera os valores entre dois intervalos definidos pelo usuário. Um mínimo e um máximo.

Exemplo:

	A	B	C	D	E	F
1	31					
2	47					
3	27					
4	19					
5	40					
6	10					
7	77					

=ALEATÓRIOENTRE(1;79)

1 – Valor mínimo do intervalo

79 – Valor máximo do intervalo

Os números aleatórios são gerados entre o valor mínimo e o máximo informados pelo usuário. É importante ressaltar que os valores serão atualizados todas as vezes que o Excel realizar um cálculo. Para fixar os valores, é preciso copiar e colar como valor, para retirar a aplicação da função.

CONVERSÃO

O Excel oferece inúmeras funções que realizam a conversão de unidades ou bases numeras, as principais funções do gênero serão apresentadas a seguir.

CONVERTER()

Esta função é extremamente útil, porém pouco conhecida dentro do Excel, a plataforma oferece a conversão de diversas unidades de medida, como gramas, libras, toneladas, newton, btu e etc. Para realizar a conversão é rápido e fácil, onde a própria ajuda suspensa do Excel sugere as unidades para conversão.

Exemplo:

	A	B	C	D	E	F	G
1	Gramas	Libras					
2	5.245	11,56325					
3	7.142	15,74541					
4	5.878	12,95877					
5	8.041	17,72737					
6	3.814	8,408431					
7	7.837	17,27763					
8	9.504	20,95273					
9	6.193	13,65323					
10	4.226	9,316735					

=CONVERTER(A2;"g";"lbm")

A2 – Célula que será convertida para outra unidade

"g" – Unidade grama origem da conversão

"lbm" – Unidade libra destino da conversão

Como citado anteriormente, a conversão é rápida e simples, basicamente se opta pela célula que será convertida, no exemplo **A2**, escolhe a sua unidade de origem ("**g**") e a sua unidade de destino após a conversão ("**lbm**").

Obs: A ajuda suspensa do Excel informa todas as unidades disponíveis para conversão, basta iniciar a digitação da fórmula.

Gramas	Libras						
5.245		=CONVERTER(A2;					
7.142		CONVERTER(núm; de_unidade; para_unidade)					
5.878	12,95877		g" - Grama				
8.041	17,72737		"sg" - Slug				
3.814	8,408431		"lbm" - Libra-massa (avoirdupois)				
7.837	17,27763		"u" - U (unidade de massa atômica)				
9.504	20,95273		"ozm" - Onça-massa (avoirdupois)				
6.193	13,65323		"grain" - Grão				
4.226	9,316735		"cwt" - Peso de 100 libras americanas (curto)				
			"uk_cwt" - Peso de 100 libras inglesas (longo)				
			"stone" - Pedras				
			"ton" - Tonelada				
			"uk_ton" - Tonelada inglesa				
			"m" - Metro				

MUDAR BASE NUMÉRICA

As funções para alterar as bases numéricas no Excel são simples e diretas, o único argumento da formula é o valor desejado ou célula que será convertida. Existe uma função para cada tipo de conversão diferente.

Lista de formulas para alterar bases decimais:

=BINADEC()	Converter números binários para decimal
=BINADEC()	Converter números binários para hexadecimal
=BINAOCT()	Converter números binários para octal
=DECABIN()	Converter números decimais para binário
=DECAHEX()	Converter números decimais para hexadecimal
=DECAOCT()	Converter números decimais para octal
=HEXABIN()	Converter números hexadecimais para binário
=HEXADEC()	Converter números hexadecimais para decimal
=OCTABIN()	Converter números octais para binário
=OCTADEC()	Converter números octais para decimais
=OCTAHEX()	Converter números octais para hexadecimal

Exemplo:

	A	B	C	D	E	F
1	Decimal	Binário				
2	7	111				
3	7	111				
4	2	10				
5	3	11				
6	4	100				
7	3	11				
8	6	110				
9	8	1000				
10	3	11				

=DECABIN(A2)

A2 – Célula que será convertida para binário

A função basicamente consiste em escolher qual a célula será convertida, existe um segundo argumento opcional para escolher quantas casas decimais serão utilizadas, porém não foi aplicável neste exemplo.

Obs: Todas as demais conversões de bases funcionam da mesma maneira, basta escolher a função desejada e aplicar na célula escolhida.

DATA E HORA

Funções relacionadas a Data e Hora, são extremamente versáteis e têm as mais variadas aplicações como por exemplo: Cálculos de banco de hora, cálculos de escala, cálculos de tempo de projeto e etc.

DIATRABALHOTOTAL()

Esta função basicamente calcula a quantidade de dias da semana entre duas datas

informadas. Existe também a possibilidade de informar feriados para serem descontados do cálculo.

Exemplo:

SOMA		X ✓ fx		=DIATRABALHOTOTAL(A2;B2;A6:A7)		
	A	B	C	D	E	F
1	Início	Final		Dias Trabalhados		
2	01/01/2018	25/02/2018		38		
3						
4						
5	Feriados					
6	01/01/2018					
7	13/02/2018					
8						

=DIATRABALHOTOTAL(A2;B2;A6:A7)

A2 - Data inicial

B2 - Data final

A6:A7 – Seleção de feriados

A fórmula é simples, consiste em escolher uma data de início, no exemplo **A2**, uma data de término que no exemplo é **B2** e o terceiro argumento é opcional, onde pode-se inserir uma seleção de células de feriados, os quais serão descontados, casos estejam dentro do intervalo solicitado, neste exemplo a região **A6:A7**.

FUNÇÕES DE HORÁRIO

As três primeiras que serão apresentadas são extremamente simples e diretas, basicamente em uma célula que contém horário, pode-se extrair apenas o seu valor de hora, minuto ou segundo, usando suas respectivas fórmulas. A última por sua vez apenas retorna a informação DO horário atual do sistema operacional.

	A	B	C
1	Exemplo	Formula	Retorno
2	04:34:20	=HORA()	4
3	04:34:20	=MINUTO()	34
4	04:34:20	=SEGUNDOS()	20
5	-	=AGORA()	00:56:02

FUNÇÕES DE DATA

As três primeiras que serão apresentadas são extremamente simples e diretas, basicamente em uma célula que contém data, pode-se extrair apenas o seu valor do ano, mês ou dia, usando suas respectivas fórmulas. A última por sua vez apenas retorna a informação da data atual do sistema operacional.

	A	B	C
1	Exemplo	Formula	Retorno
2	05/02/2018	=ANO()	2018
3	05/02/2018	=MÊS()	2
4	05/02/2018	=ANO()	5
5	-	=HOJE()	29/12/2017

Table of Contents

INTRODUÇÃO

FUNÇÕES CONDICIONAIS

SE ()

E()

OU()

SEERRO()

FUNÇÕES DE MANIPULAÇÃO DE CADEIA DE TEXTO

ESQUERDA()

EXT.TEXTO()

DIREITA()

LOCALIZAR()

NÚM.CARACT()

SUBSTITUIR ()

COMBINANDO AS FUNÇÕES

TEXTO()

ARRUMAR()

LETRAS MAIÚSCULAS

VERIFICAR CÉLULA

FUNÇÕES DE LOCALIZAÇÃO

ÍNDICE()

CORRESP()

ÍNDICE(CORRESP())

INDICE(CORRESP(); CORRESP())

PROCV()

PROCH()

PROCV(MAIOR())

DESLOC()

DESLOC(CORRESP())

INDIRETO()

CONT.VALORES()

INDICE(CONT.VALORES())

SEERRO() + FUNÇÕES DE PROCURA

FUNÇÕES DE SOMA E MÉDIA

SOMA()

SOMARPRODUTO ()

SUBTOTAL()

SOMASE()

SOMASES()

MEDIA()

MEDIASE()

OUTRAS FUNÇÕES MATEMÁTICAS

MÍNIMO ()

MÁXIMO ()

MENOR ()

MAIOR()

FATORIAL()

ARREDONADAMENTOS

ARRED()

ARREDONDAR.PARA.BAIXO()

ARREDONDAR.PARA.CIMA()

NÚMEROS ALEATÓRIOS

ALEATÓRIO()

ALEATÓRIOENTRE()

CONVERSÃO

CONVERTER()

MUDAR BASE NUMÉRICA

DATA E HORA

DIATRABALHOTOTAL()

FUNÇÕES DE HORÁRIO

FUNÇÕES DE DATA